```
x = {t, r, θ, φ}
g = DiagonalMatrix[{-(1 - (2 M) / r), (1 - 2 M / r) ^ (-1), r^2, r^2 * (Sin[θ]) ^ 2}];
Print["The Schwarzschild metric is"];
g // MatrixForm
```

Out[ ]=

{t, r, θ, φ}

The Schwarzschild metric is

Out[ ]//MatrixForm=

$$\begin{pmatrix} -1+\frac{2M}{r} & 0 & 0 & 0 \\ 0 & \frac{1}{1-\frac{2M}{r}} & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2 \operatorname{Sin}[\theta]^2 \end{pmatrix}$$

```
ig = Inverse[g];
gd = D[g, {x}];

Print["The inverse metric is"];
ig // Simplify // MatrixForm
```

The inverse metric is

Out[ ]//MatrixForm=

$$\begin{pmatrix} \frac{r}{2M-r} & 0 & 0 & 0 \\ 0 & 1-\frac{2M}{r} & 0 & 0 \\ 0 & 0 & \frac{1}{r^2} & 0 \\ 0 & 0 & 0 & \frac{\operatorname{Csc}[\theta]^2}{r^2} \end{pmatrix}$$

```
ChristoffelSymbol[g_, xx_] := Block[{ig, gd}, ig = Inverse[g];
   gd = D[g, {xx}];
   ig.(gd + Transpose[gd, {3, 1, 2}] - Transpose[gd, {2, 3, 1}]) / 2];
Print["The ChristoffelSymbol is"];
ChristoffelSymbol[g, x] // Simplify // MatrixForm
```

```
The ChristoffelSymbol is
```

Out[ ]//MatrixForm=

$$
\begin{pmatrix}
\begin{pmatrix} 0 \\ -\frac{M}{2 M r - r^2} \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} -\frac{M}{2 M r - r^2} \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} \\
\begin{pmatrix} \frac{M(-2M + r)}{r^3} \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ \frac{M}{2 M r - r^2} \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 2M - r \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ (2M - r)\,\mathrm{Sin}[\theta]^2 \end{pmatrix} \\
\begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ \frac{1}{r} \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ \frac{1}{r} \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ -(\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]) \end{pmatrix} \\
\begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ \frac{1}{r} \end{pmatrix} & \begin{pmatrix} 0 \\ 0 \\ 0 \\ \mathrm{Cot}[\theta] \end{pmatrix} & \begin{pmatrix} 0 \\ \frac{1}{r} \\ \mathrm{Cot}[\theta] \\ 0 \end{pmatrix}
\end{pmatrix}
$$

```
RiemannTensor[g_, xx_] := Block[{ig, chr, chrd, chrdif}, chr = ChristoffelSymbol[g, xx];
   chrd = D[chr, {xx}];
   chrdif = chrd - chr.chr;
   Transpose[chrdif, {4, 1, 3, 2}] - Transpose[chrdif, {4, 2, 3, 1}]];
Print["The Riemann tensor is"];
RiemannTensor[g, x] // Simplify // MatrixForm
```

```
The Riemann tensor is
```

Out[ ]//MatrixForm=

$$
\begin{pmatrix}
\begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&\frac{2M(2M-r)}{r^4}&0&0\\ \frac{2M}{(2M-r)\,r^2}&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&\frac{M(-2M+r)}{r^4}&0\\0&0&0&0\\ \frac{M}{r}&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&\cdots\\0&0&\cdots\\0&0&\cdots\\ \frac{M\,\text{Sin}[\theta]^2}{r}&0&\cdots \end{pmatrix} \\
\begin{pmatrix} 0&\frac{2M(-2M+r)}{r^4}&0&0\\ -\frac{2M}{(2M-r)\,r^2}&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&\frac{M}{(2M-r)\,r^2}&0\\0&\frac{M}{r}&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&\cdots\\0&0&\cdots\\0&0&\cdots\\0&\frac{M\,\text{Sin}[\theta]^2}{r}&\cdots \end{pmatrix} \\
\begin{pmatrix} 0&0&\frac{M(2M-r)}{r^4}&0\\0&0&0&0\\ -\frac{M}{r}&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&\frac{M}{r^2\,(-2M+r)}&0\\0&-\frac{M}{r}&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&0\\0&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&\cdots\\0&0&0&\cdots\\0&0&0&\cdots\\0&0&-\frac{2M\,\text{Si}\cdots}{\cdots}&\cdots \end{pmatrix} \\
\begin{pmatrix} 0&0&0&\frac{M(2M-r)}{r^4}\\0&0&0&0\\0&0&0&0\\ -\frac{M\,\text{Sin}[\theta]^2}{r}&0&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&0&\frac{M}{r^2\,(-2M+r)}\\0&0&0&0\\0&-\frac{M\,\text{Sin}[\theta]^2}{r}&0&0 \end{pmatrix} &
\begin{pmatrix} 0&0&0&0\\0&0&0&0\\0&0&0&-\frac{2M}{r}\\0&0&\frac{2M\,\text{Sin}[\theta]^2}{r}&0 \end{pmatrix} &
\begin{pmatrix} 0&0&\cdots\\0&0&\cdots\\0&0&\cdots\\0&0&\cdots \end{pmatrix}
\end{pmatrix}
$$

```
RicciTensor[g_, xx_] := Block[{rie}, rie = RiemannTensor[g, xx];
   Tr[Transpose[rie, {3, 2, 4, 1}], Plus, 2]];
RicciScalar[g_, xx_] := Block[{ric}, ric = RicciTensor[g, xx];
  ric.Inverse[g] // Tr]

Print["The Ricci tensor is"];
RicciTensor[g, x] // Simplify // MatrixForm

Print["The Ricci scalar is"];
RicciScalar[g, x] // Simplify
```

The Ricci tensor is

Out[•]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

The Ricci scalar is

Out[•]=

0

In[•]:=

```
M := 1;
Print["The graph of equal t-plane"];
RevolutionPlot3D[Sqrt[8 M (r - 2 M)], {r, 0, 40}, MeshStyle → {{Cyan, Opacity[1]}},
 PlotStyle → Directive[Black, Specularity[White, 40]]]
```

The graph of equal t-plane

Out[•]=